

Résumé

Dans cet article, nous donnons un algorithme qui permet de répondre très facilement au paradoxe n° 69 de la Revue d'Archimède de l'Université de Lille 1.

1 Le problème

Voici un problème paradoxal qui m'a été proposé par Aurélie Géron.

Vous êtes placé en un point O et, à une distance d'une unité, il y a un point I . Vous devez vous rendre le plus près possible d'un point X sur le segment de droite OI (par exemple, à moins d'un millièmme d'unité de X), mais vous ne pouvez vous déplacer que par sauts successifs selon la règle simple suivante :

— Vous choisissez le point O ou le point I , et votre saut vous conduit alors au point situé au milieu, entre le point P où vous êtes et le point choisi.

Si, par exemple, vous partez de O et choisissez la séquence I, O, I, I , cela vous mène respectivement aux positions

$$0,5 - 0,25 - 0,625 - 0,8125.$$

Fixons maintenant un objectif X à atteindre à moins d'un millièmme (ou un millionième, etc.). Par exemple, X situé à $\frac{1}{\pi}$ de O . Quelle est la plus courte séquence de choix O ou I qui permet de réussir ?

2 Premières réflexions

Soit n un entier naturel. Matérialisons tout d'abord la suite $(\alpha_n)_{n \in \mathbb{N}}$ définie par récurrence donnant l'abscisse du point P_n à la n^{e} étape du processus décrit plus haut (qu'on appellera « Processus du Milieu » ou PrMi en abrégé).

$$\alpha_0 = 0 \text{ ou } 1 ; \alpha_1 = \frac{1}{2} ; \forall n \geq 2, \alpha_n = \begin{cases} \frac{\alpha_{n-1}}{2}, & \text{choix } O ; \\ \frac{\alpha_{n-1}+1}{2}, & \text{choix } I. \end{cases}$$

Ainsi, géométriquement, on obtient ces ensembles de points (figure 1) :




FIGURE 1 – Ensemble des points des 4 premières itérations du processus

À la n^{e} itération du processus, les abscisses des points $P_{n,k}$ appartiennent à l'ensemble :

$$IQ(2^n) = \left\{ \frac{k}{2^n}, 1 \leq k \leq 2^n, 2 \nmid k \right\}$$




FIGURE 2 – Chemins selon le choix O/I

Que vaut $\frac{1}{\pi}$?

$$\frac{1}{\pi} \approx 0.318309886$$

Pour approcher $\frac{1}{\pi}$ avec le PrMi, on pourrait intuitivement faire de la manière suivante :

- Prenons tout d’abord le choix O , puis le choix I . On se retrouve en $P_1(1/2)$.
- On compare : $\frac{1}{\pi} < \frac{1}{2}$ donc on peut prendre le choix O pour se rapprocher de $\frac{1}{\pi}$, ce qui nous donne $P_2(1/4)$.
- $\frac{1}{4} < \frac{1}{\pi}$ donc on peut prendre le choix I pour se rapprocher de $\frac{1}{\pi}$, ce qui nous donne $P_3(5/8)$.
- $\frac{1}{\pi} < \frac{5}{8}$, donc on peut prendre le choix O pour se rapprocher de $\frac{1}{\pi}$, ce qui nous donne $P_4(5/16)$ ou $P_4(0,3125)$.
- On y est presque (on a approché $\frac{1}{\pi}$ à deux décimales près) mais on doit choisir le choix O et le choix I encore une fois, ce qui va nous éloigner de $\frac{1}{\pi}$.

La première intuition n’est pas la bonne comme le confirme la suite de l’énoncé du paradoxe :

Attention - et c’est là le côté paradoxal du problème - appliquer le principe naturel « si je suis entre O et X , je vais vers I , si je suis entre X et I , je vais vers O » ne marche pas du tout. Voici un exemple qui le prouve. Prenons l’objectif $X = 0,6$ et appliquons la méthode « si je suis entre O et X , je vais vers I , si je suis entre X et I , je vais vers O ». Rapidement, on oscille entre $\frac{1}{3}$ et $\frac{2}{3}$. Les points obtenus sont en effet (on mesure leur distance à O) :

$$0,5 - 0,75 - 0,375 - 0,6875 - 0,34375 - 0,671875 - 0,3359375 - \\ 0,66796875 - 0,333984375 - 0,6669921875 - 0,33349609375 - \\ 0,666748046875 - 0,333374023438 - 0,666687011719 - \dots$$

De plus, pour obtenir la fraction $\frac{13}{16} = 0,8125$, le chemin à prendre est même assez particulier comme le montre le schéma de la figure 2 et celle de la figure 3.

On voit, dans la figure 3, que les chemins pour arriver à une fraction du type $\frac{13}{16}$ ou encore $\frac{9}{16}$ se croisent. Nous donnons une méthode pour trouver le chemin pour arriver à $\frac{13}{16}$ dans la section suivante.

3 Chemin à suivre

Dans la figure 4, nous avons mis dans le rectangle bleu, les fractions qui ont les plus petites valeurs (« partie inférieure »), et dans le rectangle rouge, les fractions qui ont les plus grandes valeurs (« partie supérieure »). On remarque que tous les chemins de choix O nous mènent au carré bleu et tous les chemins de choix I nous mènent au chemin rouge.

On remarque aussi que tous les numérateurs de la quatrième ligne dans le rectangle bleu sont inférieurs à 7 (en fait, inférieurs à $\frac{16}{2} = 8$) et ceux de la quatrième ligne dans le rectangle rouge sont supérieurs à 9 (en fait, supérieurs à $\frac{16}{2} = 8$).




FIGURE 3 – Chemins selon la valeur des fractions




FIGURE 4 – Rangement par valeurs




FIGURE 5 – Choix des points O et I pour arriver à $\frac{13}{16}$

Pour arriver à $\frac{13}{16}$, on utilise le diagramme de la figure 5.
Ainsi, pour obtenir la fraction $\frac{13}{16}$, il faut faire les choix de points suivants :

$$[O, I, O, I, I].$$

Comment choisir entre O et I ? Par exemple, pour $\frac{a}{16}$ (avec $a \in I\mathbb{Q}(16)$), on compare a à $\frac{16}{2} = 8$. Si $a < 8$ alors il est dans le rectangle bleu donc on doit choisir le choix O (le numérateur de la fraction de l'itération précédente sera égale a et le dénominateur sera de 8), sinon il est dans le rectangle rouge donc on doit choisir le choix I (le numérateur de la fraction de l'itération précédente sera égale $a - 8$ et le dénominateur de la fraction sera égale à 8).

Et ainsi de suite...

4 Une méthode pour trouver une fraction $\frac{k}{2^n}$ proche du réel cible

Soit $r \in \mathbb{R} \cap [0, 1]$ (qu'on appellera réel cible, c'est-à-dire le réel dont on veut approcher à 10^{-p} près grâce au PrMi). On ne peut donc pas utiliser la méthode d'alternance décrite plus haut car on ne converge pas forcément vers r .

La méthode la plus simple est la suivante :

- On calcule $q_n = r \times 2^n$ pour $n \geq 1$.
- On veut ensuite trouver l'entier impair le plus proche de q . On calcule : $k_n = E(q_n)$ (où $E(q)$ est la partie entière du réel q) puis on choisit :

$$k_n^* = \begin{cases} k & \text{si } 2 \nmid k ; \\ k + 1 & \text{si } 2 \mid k . \end{cases}$$

La fraction $\frac{k_n^*}{2^n}$ conviendra si elle vérifie l'équation suivante :

$$\left| r - \frac{k_n^*}{2^n} \right| < 10^{-p}. \quad (1)$$

On peut simplifier l'équation (1) en utilisant la fonction A_p « arrondi à 10^{-p} . On la définit de la manière suivante :

Définition 1. Pour $r \in \mathbb{R} \cap [0, 1]$ s'écrivant de la façon² suivante :

$$r = \overline{0, b_1 \dots b_t \dots}^{10}$$

où les termes de la suite $(b_k)_{k \in \mathbb{N}}$ appartiennent à l'ensemble $\{0, 1, 2, 3, 4, 5, 6, 7, 8, 9\}$,

$$A_p(r) = \begin{cases} \overline{0, b_1 \dots b_{p-1} b_p}^{10} & \text{si } 0 \leq b_{p+1} < 5 ; \\ \overline{0, b_1 \dots b_{p-1} c_p}^{10} & \text{si } 5 \leq b_{p+1} \leq 9 ; \end{cases}$$

avec $c_p = b_p + 1$.

Avec cette fonction ainsi définie pour $r \in \mathbb{R} \cap [0, 1]$, l'équation (1) peut se traduire par :

$$A_p(r) = A_p\left(\frac{k_n^*}{2^n}\right)$$

Remarque 1. Sur AlgoBox (logiciel utilisé pour faire l'algorithme de la section 5), pour arrondir un réel r à 10^{-p} près (p étant un entier) :

1. Au tout début, on pourrait prendre une infinité de fois le choix O (respectivement le choix I), cela nous menerait toujours au point O (respectivement le choix O). Quitte à faire cela, au tout début du processus, nous choisirons les choix O et I .
2. La notation $r = \overline{XYZ}^{10}$ représente l'écriture décimale (ou en base 10) du nombre r

```

Code de l'algorithme
VARIABLES
  r EST_DU_TYPE NOMBRE
  p EST_DU_TYPE NOMBRE
DEBUT_ALGORITHME
  LIRE r
  LIRE p
  AFFICHER r
  r PREND_LA_VALEUR r*pow(10,p)
  r PREND_LA_VALEUR round(r)
  r PREND_LA_VALEUR r/pow(10,p)
  AFFICHER r
FIN_ALGORITHME

```

(où $\text{round}(r)$ permet d'arrondir r à l'entier le plus proche et $\text{pow}(10, p) = 10^p$). On peut aussi tronquer un réel r à 10^{-p} près avec le code suivant :

```

Code de l'algorithme
VARIABLES
  r EST_DU_TYPE NOMBRE
  p EST_DU_TYPE NOMBRE
DEBUT_ALGORITHME
  LIRE r
  LIRE p
  AFFICHER r
  r PREND_LA_VALEUR r*pow(10,p)
  r PREND_LA_VALEUR floor(r)
  r PREND_LA_VALEUR r/pow(10,p)
  AFFICHER r
FIN_ALGORITHME

```

(où $\text{floor}(r)$ permet d'obtenir la partie entière $E(r)$ de r).

5 L'algorithme

Soit $r \in \mathbb{R} \cap [0, 1]$ notre réel cible et $p \in \mathbb{N}$. On veut approcher r à 10^{-p} par le PrMi. L'algorithme suivant permet de faire cela, il se décompose en trois parties :

1ère partie : de la ligne 1 à la ligne 21, l'utilisateur choisit la valeur du réel cible r (qui doit être compris entre 0 et 1 inclus) et la précision 10^{-p} (en fait, il choisit la variable p). Le programme teste la validité des variables r et p , c'est-à-dire il vérifie si $0 \leq r \leq 1$ et $p \in \mathbb{N}$.

2ème partie : de la ligne 22 à la ligne 53, le programme teste les valeurs k_n^* (méthode décrite dans la section précédente) jusqu'à tant que l'équation (1) soit vérifiée. Il affiche ensuite k_n^* et 2^n .

3ème partie : de la ligne 54 à la ligne 75, le programme essaie de trouver le chemin (choix O/I) qui permet d'arriver à la fraction $\frac{k_n^*}{2^n}$ qui convient dans l'approximation du réel cible à 10^{-p} près. La méthode a été décrite à la section 3. Le résultat final est une série de 0 et de 1 (stocké dans une liste), le O représentant le choix O et le 1, le choix I .

```

Code de l'algorithme
1  VARIABLES
2    r EST_DU_TYPE NOMBRE
3    p EST_DU_TYPE NOMBRE
4    s EST_DU_TYPE NOMBRE
5    t EST_DU_TYPE NOMBRE
6    m EST_DU_TYPE NOMBRE

```

```

7     n EST_DU_TYPE NOMBRE
8     k EST_DU_TYPE NOMBRE
9     e EST_DU_TYPE NOMBRE
10    y EST_DU_TYPE NOMBRE
11    z EST_DU_TYPE NOMBRE
12    L EST_DU_TYPE LISTE
13    DEBUT_ALGORITHMME
14    AFFICHER "r? (0 <= r <= 1) "
15    LIRE r
16    AFFICHER "Précision à 10^{-p} ; p?"
17    LIRE p
18    SI (r>1 ou p-floor(p)!=0) ALORS
19        DEBUT_SI
20        AFFICHER "Variables r et p non conformes !"
21        FIN_SI
22    SINON
23        DEBUT_SINON
24        s PREND_LA_VALEUR r*pow(10,p)
25        s PREND_LA_VALEUR round(s)
26        s PREND_LA_VALEUR s/pow(10,p)
27        n PREND_LA_VALEUR 0
28        k PREND_LA_VALEUR r*pow(2,n)
29        e PREND_LA_VALEUR floor(k)
30        SI (e%2==0) ALORS
31            DEBUT_SI
32            e PREND_LA_VALEUR e+1
33            FIN_SI
34        t PREND_LA_VALEUR e/(pow(2,n))*pow(10,p)
35        t PREND_LA_VALEUR round(t)
36        t PREND_LA_VALEUR t/pow(10,p)
37        TANT_QUE (t != s) FAIRE
38            DEBUT_TANT_QUE
39            n PREND_LA_VALEUR n+1
40            k PREND_LA_VALEUR r*pow(2,n)
41            e PREND_LA_VALEUR floor(k)
42            SI (e%2==0) ALORS
43                DEBUT_SI
44                e PREND_LA_VALEUR e+1
45                FIN_SI
46            t PREND_LA_VALEUR e/(pow(2,n))*pow(10,p)
47            t PREND_LA_VALEUR round(t)
48            t PREND_LA_VALEUR t/pow(10,p)
49            FIN_TANT_QUE
50        z PREND_LA_VALEUR pow(2,n)
51        AFFICHER e
52        AFFICHER z
53        FIN_SINON
54    y PREND_LA_VALEUR n+1
55    TANT_QUE (n>1) FAIRE
56        DEBUT_TANT_QUE
57        SI (e>pow(2,n-1)) ALORS
58            DEBUT_SI
59            e PREND_LA_VALEUR e-pow(2,n-1)
60            L[n+1] PREND_LA_VALEUR 1

```

```

61     FIN_SI
62     SINON
63         DEBUT_SINON
64         L[n+1] PREND_LA_VALEUR 0
65         FIN_SINON
66     n PREND_LA_VALEUR n-1
67     FIN_TANT_QUE
68     L[2] PREND_LA_VALEUR 1
69     L[1] PREND_LA_VALEUR 0
70     POUR m ALLANT_DE 1 A y
71         DEBUT_POUR
72             AFFICHER L[m]
73             AFFICHER " "
74         FIN_POUR
75     FIN_ALGORITHME

```

6 Exemples

Dans cet exemple, on donne des exemples qui répondent à la question posée dans le paradoxe n° 69 de la Revue d'Archimède de l'Université de Lille I.

6.1 $r = \frac{1}{\pi}$, $p = 3$

```

***Algorithme lancé***
r? (0 <= r <= 1)
Entrer r : 1/3.14159265359
Précision à 10^{-p} ; p?
Entrer p : 3
163
512
0 1 1 0 0 0 1 0 1 0
***Algorithme terminé***

```

On peut vérifier que :

$$\frac{1}{\pi} \approx \underline{0.31830988618} \quad \text{et} \quad \frac{163}{512} \approx \underline{0.318359375}$$

et que le chemin $[O, I, I, O, O, O, I, O, I, O]$ est celui qui nous mène à la fraction $\frac{163}{512}$ (voir figure 6).

6.2 $r = \frac{1}{\pi}$, $p = 6$

```

***Algorithme lancé***
r? (0 <= r <= 1)
Entrer r : 1/3.14159265359
Précision à 10^{-p} ; p?
Entrer p : 6
83443
262144
0 1 1 0 0 1 1 1 1 1 0 1 0 0 0 1 0 1 0
***Algorithme terminé***

```

On peut vérifier que

$$\frac{1}{\pi} \approx \underline{0.31830988618} \quad \text{et} \quad \frac{83443}{262144} \approx \underline{0.31830978393}$$




FIGURE 6 – Chemin vers $\frac{163}{512}$

et on laissera aux lecteurs le soin de vérifier que le chemin $[O, I, I, O, O, I, I, I, I, O, I, O, O, O, I, O, I, O]$ est le chemin qui mène vers la fraction $\frac{83443}{262144}$.

6.3 $r = 0,6$, $p = 7$

Dans la section 2, nous avons vu que le chemin $[O, I, O, I, O, I, \dots]$ ne permet pas d'atteindre $r = 0,6$. Qu'en est-il ?

Bien que $0,6$ peut s'écrire simplement $\frac{6}{10}$, la fraction $\frac{6}{10}$ ne peut pas s'écrire (simplement) comme $\frac{k_n^*}{2^n}$. On a choisi, ici, une approximation de 10^{-7} pour approximer $r = 0,6$ avec le PrMi.

```

***Algorithme lancé***
r? (0 <= r <= 1)x
Entrer r : 0.6
Précision à 10^{-p} ; p?
Entrer p : 7
5033165
8388608
0 1 0 1 1 0 0 1 1 0 0 1 1 0 0 1 1 0 0 1 1 0 0 1
***Algorithme terminé***

```

On a :

$$\frac{5033165}{8388608} = 0,600000023842$$

et que le chemin à choisir pour arriver à $\frac{5033165}{8388608}$ est une alternance de choix $I/I/O/O$ à partir de la troisième itération.